INTERACTIVE SYSTEMS

- Machine repairman model
- Multiprocessor interactive systems

Jozo Dujmović

INTERACTIVE SYSTEMS





S = service time

z = think time (5-50 sec)

$\lambda = 1/Z$

$\mu = 1/S$



"MACHINE REPAIRMAN MODEL"

j = STATE OF THE SYSTEM



RESPONSE TIME FORMULA



Assumptions:

(1) All workstations have the same priority

(2) All users generate the same workload (ATM)

Consequence:

During the interaction cycle time Z + R the central server (CP) completes n transactions (one transaction per each of n workstations)

$$X = \frac{n}{Z+R} = \frac{U_p}{S}$$

$$U_p = \frac{nS}{Z+R}$$

$$\boxed{R = nS/U_p - Z}$$

$$\boxed{\rho = \lambda / \mu} = S/Z = \frac{\text{SERVICE TIME}}{\text{THINK TIME}}$$

$$p_1 = 4\rho\, p_0$$

$$p_2 = 3\rho\, p_1 = 12\rho^2 p_0$$

$$p_3 = 2\rho\, p_2 = 24\rho^3 p_0$$

$$p_4 = \rho\, p_3 = 24\rho^4 p_0$$

Balance Equations

$$p_0 + p_1 + p_2 + p_3 + p_4 = 1$$

$$p_0 (1 + 4\rho + 12\rho^2 + 24\rho^3 + 24\rho^4) = 1$$

$$p_0 = \frac{1}{1 + 4\rho + 12\rho^2 + 24\rho^3 + 24\rho^4}$$

$$U_p = 1 - p_0$$

$$= \frac{4\rho + 12\rho^2 + 24\rho^3 + 24\rho^4}{1 + 4\rho + 12\rho^2 + 24\rho^3 + 24\rho^4}$$

CRITICAL NUMBER OF USERS n^*

$n = 1$: $\underline{R = S}$, $X = 1/(Z+S)$, $U_p = S/(Z+S)$

$n \gg 1$: $\underline{U_p \to 1}$ (saturation)

$R(n) = nS - Z$ (asymptote)



$$n^* S - Z = S \quad \Rightarrow \quad \boxed{n^* = \frac{S+Z}{S}}$$

n^* is the main parameter of an interactive system; it denotes the beginning of saturation

$n \le n^*$: normal operation

$n > n^*$: saturation

Example

The ATM transaction generates processor activity of 0.1 sec. User's think time is 4 sec. Utilization of ATM is not greater than 50%.

(a) Estimate the number of machines that the bank can install.

(b) What is the maximum number of machines if the response time must be less than the user's think time

(c) Analyze the response time $R(n)$ and CP utilization

(a) $n^* = \frac{S+Z}{S} = \frac{4+0.1}{0.1} = 41$

If no saturation effect are allowed the maximum number of machines should be

$$N \cong 2n^* = 82$$

(b) If $R = 4$ sec the central subsystem is saturated:

$$R = NS - Z \leq Z$$

$$N \leq 2Z/S = 80$$

(c)

n	20	40	60	80
$R(n)$	0.178	0.526	2.004	4.000
$U_p(n)$	0.479	0.884	0.999	1.000





```
C==============================================================
C     CLOSED QUEUEING MODEL OF AN INTERACTIVE SYSTEM
C     Jozo J. Dujmovic, 1985
C==============================================================
      SUBROUTINE IS(THINK,SERVE,NTERM,PZERO,CRITN,PQLEN,TRESP)
C   I n p u t s:
C     THINK = average think time per interaction   (THINK >= 0)
C     SERVE = average service time per interaction (SERVE >  0)
C     NTERM = number of active terminals           (NTERM >= 1)
C   O u t p u t s:
C     PZERO = processor idle time
C     CRITN = critical number of terminals
C     PQLEN = average length of processor queue
C     TRESP = average response time (same units as THINK,SERVE)
C--------------------------------------------------------------
      CRITN = 1. + THINK / SERVE
      IF (THINK .GT. 0.) THEN
          RHO   = SERVE/THINK
          SUM   = 1.
          DO J = 1,NTERM
             IF (SUM .LT. 1.E30) SUM = 1. + J*RHO*SUM
          END DO
          PZERO = 1./SUM
          IF (PZERO .GT. 0.0001) THEN
             PLOG  = 0.
             PQLEN = 0.
             DO J = 1,NTERM
                PLOG  = PLOG + ALOG(NTERM+1.-J) + ALOG(RHO)
                PJLOG = PLOG + ALOG(FLOAT(J))
                IF (PJLOG .GE. -85.) PQLEN = PQLEN + EXP(PJLOG)
             END DO
             PQLEN = PQLEN * PZERO
          ELSE
             PQLEN = NTERM - THINK/SERVE
          END IF
      ELSE
          PZERO = 0.
          PQLEN = NTERM
      END IF
      TRESP = NTERM * SERVE / (1.-PZERO) - THINK
      RETURN
      END
```

```
THINK, SERVE, NTERM = 4 0.1 1

PZERO =     .976      UTILI =     .024
CRITN =   41.000      PQLEN =     .024
TRESP =     .100      QWAIT =     .000
CYCLE =    4.100
RNORM =    1.000      TNORM =    1.000

THINK, SERVE, NTERM = 4 0.1 20

PZERO =     .521      UTILI =     .479
CRITN =   41.000      PQLEN =     .852
TRESP =     .178      QWAIT =     .078
CYCLE =    4.178
RNORM =    1.780      TNORM =    1.019

THINK, SERVE, NTERM = 4 0.1 40

PZERO =     .116      UTILI =     .884
CRITN =   41.000      PQLEN =    4.646
TRESP =     .526      QWAIT =     .426
CYCLE =    4.526
RNORM =    5.257      TNORM =    1.104

THINK, SERVE, NTERM = 4 0.1 60


PZERO =     .001      UTILI =     .999
CRITN =   41.000      PQLEN =   20.027
TRESP =    2.004      QWAIT =    1.904
CYCLE =    6.004
RNORM =   20.041      TNORM =    1.464

THINK, SERVE, NTERM = 4 0.1 80

PZERO =     .000      UTILI =    1.000
CRITN =   41.000      PQLEN =   40.000
TRESP =    4.000      QWAIT =    3.900
CYCLE =    8.000
RNORM =   40.000      TNORM =    1.951

THINK, SERVE, NTERM = 4 0.1 100

PZERO =     .000      UTILI =    1.000
CRITN =   41.000      PQLEN =   60.000
TRESP =    6.000      QWAIT =    5.900
CYCLE =   10.000
RNORM =   60.000      TNORM =    2.439

THINK, SERVE, NTERM =
```

T_0 = matrix inversion time in monoprogrammed environment without active terminals

T_5 = matrix inversion time if 5 high-priority terminals are active during matrix inversion.

Matrix inversion is a low-priority batch job. Compute T_5 if

$T_0 = 10$ min

$Z = 10$ sec (think time)

$S = 1$ sec (service time)





$$\rho = \frac{\lambda}{\mu} = \frac{5}{Z} = \frac{1}{10} = 0.1$$

$$p_1 = 5\rho p_0$$
$$p_2 = 4\rho p_1$$
$$p_3 = 3\rho p_2$$
$$p_4 = 2\rho p_3$$
$$p_5 = \rho p_4$$
$$p_0 + p_1 + p_2 + p_3 + p_4 + p_5 = 1$$

$$\frac{1}{p_0} = \sum_{j=0}^{5} \frac{5!}{(5-j)!}\rho^j = 1 + 5\rho + 20\rho^2 + 60\rho^3 + 120\rho^4 + 120\rho^5 = 1.7732$$

Matrix inversion is active only if interactive workload is NOT ACTIVE.

$$T_0 = T_5 \cdot p_0$$

$$T_5 = \frac{T_0}{p_0} = \underline{\underline{17.732 \text{ min}}}$$

$$U = 1 - p_0 = 0.436$$

(processor utilization by the interactive workload)

MULTIPROCESSOR INTERACTIVE SYSTEMS

n = # of terminals/workstations

m = # of processors

Z = average think time ; $\lambda = 1/Z$

S = average service time; $\mu = 1/S$





Processor utilization / idling

$$1 - U = p_0 + \frac{m-1}{m} p_1 + \frac{m-2}{m} p_2 + \cdots + \frac{1}{m} p_{m-1}$$

$$U = 1 - \sum_{i=0}^{m-1} \frac{m-i}{m} p_i$$

Balance equations

$$\rho = \frac{\lambda}{\mu} = \frac{S}{Z}$$

$$p_1 = n\rho p_0$$

$$p_2 = \frac{n-1}{2}\rho p_1 = \frac{n(n-1)}{2}\rho^2 p_0$$

$$\vdots$$

$$p_k = \frac{n+1-k}{\min(m,k)} p_{k-1}\rho = \prod_{j=1}^{k} \frac{n+1-j}{\min(m,j)} \cdot \rho^k p_0$$

$$\vdots$$

$$p_0 + p_1 + \cdots + p_n = 1$$

$$p_0\left(1+\sum_{k=1}^{n}\prod_{j=1}^{k}\frac{n+1-j}{\min(m,j)}\rho^k\right)$$

$$p_0 = \frac{1}{1+\sum_{k=1}^{n}\prod_{j=1}^{k}\frac{n+1-j}{\min(m,j)}\rho^k}$$

$$p_k = \frac{\prod_{j=1}^{k}\frac{n+1-j}{\min(m,j)}\rho^k}{1+\sum_{k=1}^{n}\prod_{j=1}^{k}\frac{n+1-j}{\min(m,j)}\rho^k}, \quad k>0$$

Response time

Interaction cycle time = R+Z

During each cycle all n workstations must be served (they have equal priority)

Consumed processor time = nS

Delivered processor time = $m(R+Z)U$

Consumed time must be equal to delivered time, yielding the response time equation:

$$(R+Z)\,mU = nS$$

$$\boxed{\begin{aligned} R &= \frac{nS}{mU} - Z \\ &= S\ , \quad n=1 \\ &= nS/m - Z\ , \quad n \gg 1 \end{aligned}}$$

Throughput

$$X = \frac{n}{R+Z}$$

Critical number of users n^*



$$n^* S / m - Z = S$$

$$\boxed{n^* = m \frac{Z+S}{S}}$$

EXAMPLE

$Z = 8$ sec $\quad S = 2$ sec $\quad m = 2 \quad \rightarrow \quad n^* = 10$

$Z = 8$ sec $\quad S = 2$ sec $\quad m = 1 \quad \rightarrow \quad n^* = 5$

```
C=================================================================
C      CLOSED QUEUEING MODEL OF A MULTIPROCESSOR INTERACTIVE SYSTEM
C      This is a straightforward model without special compensation
C      of overflow/underflow errors
C      Jozo J. Dujmovic, 1996
C=================================================================
      SUBROUTINE MPIS(THINK,SERVE,NTERM,NPROC,  U,CRITN,PQLEN,R,X)
C   I n p u t s:
C      THINK = average think time per interaction
C      SERVE = average service time per interaction
C      NTERM = number of active terminals             (NTERM >= 1)
C      NPROC = number of processors                   (NPROC >= 1)
C   O u t p u t s:
C      U     = processor utilization
C      CRITN = critical number of terminals
C      PQLEN = average length of processor queue
C      R     = average response time (same units as THINK,SERVE)
C      X     = throughput (transactions per time unit)
C-----------------------------------------------------------------
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION F(0:500), P(0:500)
      CRITN = NPROC * (SERVE + THINK) / SERVE
      RHO   = SERVE/THINK
      F(0)  = 1.
      SUM   = 1.
      DO J = 1, NTERM
         F(J)  = F(J-1)*(NTERM+1.-J)*RHO/min(NPROC, J)
         SUM   = SUM + F(J)
      END DO
      DO J = 0, NTERM
         P(J)  = F(J)/SUM
      END DO
      PIDLE = 0.
      DO I = 0, NPROC-1
         PIDLE = PIDLE + P(I)*FLOAT(NPROC-I)/FLOAT(NPROC)
      END DO
      U     = 1. - PIDLE
      R     = NTERM*SERVE/(U*NPROC) - THINK
      X     = NTERM/(THINK+R)
      PQLEN = R*X
      END

C===============================================================
C     MULTIPROCESSOR INTERACTIVE SYSTEM PERFORMANCE COMPARISON
C===============================================================
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION THINK(10),SERVE(10),NPROC(10), R(10), X(10), U(10)
      WRITE(*,'(/'' THE NUMBER OF CASES IN THIS ANALYSIS = ''\)')
      READ(*,*) K
      WRITE(*,'('' THE MAXIMUM NUMBER OF TERMINALS = ''\)')
      READ(*,*) N
      WRITE(*,'('' OUTPUT ( 1=R , 2=X , 3=U ) = ''\)')
      READ(*,*) M
      DO I=1,K
        WRITE(*,'(/'' Case '',I2,'' :      THINK, SERVE, NPROC = ''\)')I
        READ(*,*) THINK(I), SERVE(I), NPROC(I)
      END DO
      WRITE(*,*)
      DO NTERM=1,N
         DO I=1,K
            CALL MPIS(THINK(I), SERVE(I), NTERM, NPROC(I),
     ,                U(I), CRITN, PQLEN, R(I), X(I))
         END DO
         IF(M .EQ. 1) WRITE(*,'(I3,10E12.5)') NTERM, (R(I), I=1,K)
         IF(M .EQ. 2) WRITE(*,'(I3,10E12.5)') NTERM, (X(I), I=1,K)
         IF(M .EQ. 3) WRITE(*,'(I3,10E12.5)') NTERM, (U(I), I=1,K)
      END DO
      END
```

$Z = 8$
$S = 2$
$n_1^* = 5$
$n_2^* = 10$



$Z=8$
$S=2$
$n_1^*=5$
$n_2^*=10$